FIRST COMPUTER LIBRARY

COMPUTER FUN



USBORNE

FIRST COMPUTER LIBRARY

COMPUTER FUN

Gaby Waters
Illustrated by Graham Round

Additional programs by Paul Shreeve
Series Editor: Helen Davies

Contents

About this book

This book is full of fun programs for you to type in and run on a computer. There are also ideas for changing the programs and extra lines you can add to make them more exciting. At the back of the book there are some instructions for adding sound and colour too.



You do not need to know anything about programming to run the programs. You can find out how to type them in and make them work on the next few pages. There are also lots of hints to help you if things go wrong.





The programs are written in a computer language called BASIC. Each home computer understands a slightly different version of BASIC and you will need to change some program lines to suit your computer. These lines are clearly marked and on pages 40-46 there are lists of line changes for Spectrum, BBC, Electron, Commodore 64, VIC 20 and Apple computers.

If you have a different make of computer, you may be able to work out the changes you need to make from your computer's manual.

Typing in programs

To make a computer do something, you need to type in a list of instructions called a program. Programs for home computers are written in BASIC. BASIC is quite like English, with some special words and symbols. Below you can see what a program written in BASIC looks like.



This program will make a computer ask your name and then put a message on the screen. Each instruction in the program is on a separate line and each line begins with a number. The numbers make the computer carry out the instructions in the right order.

How to type in a program

When you switch on a computer, a symbol called a cursor flashes on the screen. This means that you can start typing in a program.

At the end of each line, press the key marked RETURN, or ENTER. This puts the program line into the computer's memory.

Type the program just as it is printed. On most computers you type BASIC commands letter by letter, but on some you just press one key.

If you make a mistake, rub it out by pressing the delete key. The label on the key varies from one make of computer to another.

Running the program

When you have finished typing in the program, you type the command RUN and press RETURN (or ENTER). This is called running the program and it makes the computer carry out your instructions. RUN does not need a line number because it is not part of the program.

If you typed in the program on the opposite page, try running it. First the computer will ask your name. Type it in and press RETURN (or ENTER). Then the computer will give you a message. You can make the computer run the program again by typing RUN again and pressing RETURN (or ENTER). Try it a few more times, using different names.

Getting rid of the program

When you have finished with the program, just type NEW and press RETURN (or ENTER). This wipes the instructions from the computer's memory so it is ready for you to type in another program.

What to do if things go wrong

If a program does not work you have probably made a mistake typing it in. It is very easy to make mistakes when you type in programs. Mistakes in programs are called bugs. Some computers let you know if there is a bug in a program line when you press RETURN or ENTER, but most wait until you try to run the program.



The computer tells you there is a bug in a program by putting a special message on the screen like the one above. This is called an error message. It usually tells you what sort of mistake it is and what line it is in. Error messages are different on different computers and are often written in a sort of code. To find out what they mean, look in your computer's manual.

How to de-bug a program

First type LIST and press RETURN* to display the program lines on the screen. Then check through the program lines for mistakes.

Bugs are not easy to find, but it is easier to spot them if you know what to look for. The bug spotting guide on page 47 gives you some handy tips and hints.

When you spot the bug, type the whole line again and press RETURN. Now if you list the program again, you will see that the new line has replaced the old one.

*If your computer's key is marked ENTER, press it each time you see RETURN from now on.

Changing a program

In this book there are lots of ideas for adapting programs by adding or changing program lines. You do this in a similar way to de-bugging, as shown below.

First type LIST and press RETURN to put the program lines on the screen. To change a line, type in the line number and the new instruction and press RETURN.



You add extra lines to programs in the same way. Their line numbers tell the computer where to put them in the program.

Special words

Most BASIC commands are understood by all computers but there are some which vary from computer to computer. In this book, lines with commands which vary are printed on a red stripe. In the boxes below you can find out what to do when you come across one of these.

When you see a line printed on a red stripe, turn to the conversion pages at the back of the book. Find the page for your computer, then look for the name and page number of the program you are typing in.





If the line needs changing for your computer, you will find a conversion line with the same line number. Type this line instead of the one on the stripe. If there is no conversion line, this line does not need changing for your computer and you can type the line on the stripe, just as it is.

Happy Birthday

The program on this birthday card makes the computer put a special birthday message on the screen. You could run it when one of your friends or family has a birthday. To see how the program works, type it into your computer and run it.



Running the program

When you run the program, the computer will write the words HAPPY BIRTHDAY letter by letter on the screen and then LOVE FROM YOUR COMPUTER.

Changing the message



You can change the birthday message by typing different words inside the quotes in lines 20, 90 and 100. To do this, type LIST and press RETURN to put the program lines on the screen. Then retype the lines you want to change. In the pictures above there are some ideas for changing the message or you can invent messages of your own.

Making the message repeat



Now try adding the lines above to the program. They make the computer print the message again and again until you stop the program, or switch off the computer. You can find out how to stop the program on the right.

Stopping the program

You can make the computer stop running a program by pressing the ESCAPE key. The label on this key varies on different computers. You can find out what your computer calls it by looking on page 46 of this book or in your computer's manual.

Space monsters

This program makes a picture of a computer space monster on your screen. The monster shape is made up of X's, so you must be careful to type all the X's and the spaces in the right place. The design on the paper below will help you to count the number of X's and spaces in each line. In the program the spaces are marked with the symbol □. For each □, press the space bar once.



Space monster program

```
10 CLS
20 PRINT:PRINT:PRINT
30 PRINT "□□□□□X□X"
40 PRINT "□□□□XXXXX"
50 PRINT "□□□□X□X□X"
60 PRINT "□□□□XXXXX"
70 PRINT "XXXXXXXXXXXXX"
80 PRINT "XX□□XXXXX□□XX"
90 PRINT "XX□□□X□X□□□XX"
100 PRINT "□□□□XX□XX"
```

Remember to check line 10 in the conversion pages.

There are 5 spaces here.

There are 13 X's here.

Use this drawing of the monster to help you count the X's and spaces.

The monster may look smaller on your screen.

Running the program

Type RUN to make the program work. If your monster looks wrong, type LIST and press RETURN and check the program lines to make sure you have the right number of X's and spaces.

Making the monster move

If you add the program lines printed below you can make the monster wave its arms up and down. To do this, first type LIST followed by RETURN to put the program lines on the screen, then type in the extra lines. Make sure you type each line exactly as it is printed and press the space bar every time you see the symbol □.



```
110 FOR K=1 TO 300:NEXT K
120 CLS
130 PRINT:PRINT:PRINT
140 PRINT "□□□□□X□X"
150 PRINT "□□□□XXXXX"
160 PRINT "XX□□X□X□X□□XX"
170 PRINT "XX□□XXXXX□□XX"
180 PRINT "XXXXXXXXXXXXX"
190 PRINT "□□□□XXXXX"

200 PRINT "□□□□□X□X"
210 PRINT "□□□□XX□XX"
220 FOR K=1 TO 300:NEXT K
230 GOTO 10
```

There are 13 X's here.

There are 5 spaces here.

Design your own monster

1 Try making the monster look different by using letters such as O or S instead of X's. To do this, retype all the lines starting with PRINT, using your letter instead of the X's.

2



If you like, you can make a new monster of your own. Design the monster on a piece of squared paper like this. Then type NEW and press RETURN. Type the program again putting the letters and spaces from your design in quotes in PRINT lines.

Red Alert

The program on this page puts a flashing "Red Alert" message on the screen. Try typing it into your computer.

```
10 CLS
20 FOR K=1 TO 300:NEXT K
30 PRINT:PRINT:PRINT
40 PRINT TAB(3);"DANGER"
50 PRINT TAB(3);"RED ALERT"
60 FOR K=1 TO 300:NEXT K
70 GOTO 10
```

Look at the conversion pages to see if you need to change this line.



The message will carry on flashing until you stop the program.

Running the program

Type RUN to start the program working. To stop the program, press ESCAPE (or your computer's key).

Changing the program

In the boxes below there are some ideas for changing the program. To change the program, stop it running and type LIST. Then type in the new versions of the lines you want to change.

1

Type this in lines 20 and 60 to make the message flash faster.

```
FOR K=1 TO 150:NEXT K
```

```
FOR K=1 TO 500:NEXT K
```

Type this in lines 20 and 60 to make the message flash more slowly.

You can make the message flash at a different speed by changing the number 300 in lines 20 and 60. A smaller number makes it flash quickly and a larger number makes it flash slowly.

2

```
40 PRINT TAB(14);"DANGER"
50 PRINT TAB(8);"RED ALERT"
```

Try these numbers.

You can move the message further across the screen by putting a bigger number in brackets after TAB in lines 40 and 50. The number tells the computer how many spaces to leave before the words on the screen.

More things to do

Try putting a new message, like the one above, inside the quotes in lines 40 and 50.

Next time you go into a computer shop try typing the program, with a message like this, into one of the shop's computers.

Personal messages

You can make the computer send a flashing message addressed to you, or whoever is running the program, by adding the lines on the right. To do this, stop the program, type LIST and then type in the new lines. You can find out how the new program works below.

```
 2 CLS
 4 PRINT "WHAT IS YOUR NAME?"
 6 INPUT N$
40 PRINT TAB(3);N$;" IS GREAT"
50 PRINT TAB(3);N$;" RULES O.K."
```

First the computer asks you your name. You should type it in and press RETURN.

Then the computer flashes these messages with your name in them on the screen.

Try the program on your friends or make up some other names to type in.

Drawing pictures

This program lets you make patterns and simple pictures with lines of stars on the screen. You tell the computer where to draw the lines by pressing certain keys. First type in the program, then look at the bottom of the page to find out how to use it.

Picture program

```
10 CLS
20 LET X=10:LET Y=10
30 LET A$=INKEY$
40 IF A$="" THEN GOTO 30
50 IF A$="A" THEN LET Y=Y-1
60 IF A$="Z" THEN LET Y=Y+1
70 IF A$="N" THEN LET X=X-1
80 IF A$="M" THEN LET X=X+1
90 PRINT TAB(X,Y);"* "
100 GOTO 30
```

There is no space between the quote marks in this line.

Look at the conversion pages to see if you need to change any of the lines on red stripes.

How to use the program

1



When you type RUN the screen will go blank. To start drawing, press any key. A star will appear on the screen.

2



To draw a line of stars going up the screen, press A. To draw a line going down, press Z. To draw lines left and right press N and M.

3



Try making patterns on your screen. To stop the program, press the key marked ESCAPE (or your computer's key).

* If this happens, press ESCAPE (or your computer's key) and start again.

Rubbing out stars



These extra program lines enable you to rub out the last star you put on the screen by pressing the space bar. Before you type in the extra lines, stop the program and type LIST to put the program on the screen.

Printing spaces



You can make a row of spaces by pressing one of the direction keys, then the space bar, several times. This is very useful for drawing pictures that need breaks in the lines.

Patterns with different symbols



You can draw patterns with different symbols, such as & or #. To do this, list the program again and type in these new lines. Now when you run the program, the computer starts by asking you what symbol you want to use, as shown on the screen above.

Drawing in colour

Computer Snap

You can make your computer play Snap by typing in the program on the right. Remember to check the conversion pages to see whether you need to change any of the lines on the red stripes. You can find out how to play the game opposite.

Leave a space inside the quotes before the words WELL DONE.

Snap program

Type two spaces inside the quotes in line 50.

```
10 CLS
20 LET N=0:PRINT:PRINT:PRINT
30 LET X=INT(RND(1)*10+1)
40 LET Y=INT(RND(1)*10+1)
50 PRINT X;"    ";Y;
60 LET A$=INKEY$
70 LET N=N+1
80 IF A$="S" AND X=Y THEN GOTO 110
90 IF N<=100 THEN GOTO 60
100 GOTO 10
110 PRINT "  WELL DONE"
120 FOR K=1 TO 1000:NEXT K
130 GOTO 10
```

Changing speed

You can change the speed of the game very easily by altering the number 100 in line 90. A bigger number makes the game slower and a smaller number makes it faster. Try changing line 90 as shown in the boxes on the right. If the game is still not the right speed for you, try other numbers in line 90.

1

This line makes the game slower.

```
90 IF N<=200 THEN GOTO 60
```

These are the numbers which alter the speed.

2

```
90 IF N<=50 THEN GOTO 60
```

This line makes the game faster.

How to play the game



When you run the program, the computer puts a pair of numbers on the screen as shown in the picture above.

Then it puts another pair on the screen, and another, and so on, changing the numbers each time.

When the two numbers are the same, try and press the S key (for Snap) before the new numbers appear on the screen.

Word and number snap



You can change the game so that the computer writes the first number as a word. To do this, add the program lines in the box above. First stop the program, then list it and type in the new lines. If you find the new game is too fast, you can change the speed again to make it slower.

Code-writer

Here is a program to make a computer code and decode messages. You can use it to send top-secret information, or write private letters to your friends. Nobody will be able to crack the code without this program and the line changes shown opposite.

Code program

```
10 CLS
20 LET C$="":PRINT
30 PRINT "TYPE IN YOUR MESSAGE"
40 INPUT M$:LET L=LEN(M$)
50 IF INT(L/2)<>L/2 THEN LET M$=M$+" "
60 FOR K=1 TO LEN(M$) STEP 2
70 LET C$=C$+MID$(M$,K+1,1)
80 LET C$=C$+MID$(M$,K,1)
90 NEXT K
100 CLS
110 PRINT:PRINT "CODE MESSAGE:"
120 PRINT:PRINT C$
```

Leave no space inside these quotation marks.

Type one space inside these quotation marks.

How to use the program



When you run the program, the computer asks you to type in your message.

You should type it in and then press RETURN.

The computer translates your message into code and puts it on the screen.

Exchanging secret messages



If any of your friends have a computer you can exchange messages in code. You can pass them to one another by leaving them in a secret

hiding place. Spies call this a dead letter box. Your friend can pick up your messages when the coast is clear.

To read the messages you have to decode them back into English. You can find out how to change the program to do this below.

Decoding messages

To decode secret messages you need to change lines 30 and 110 of the code program. List the program and type in the new lines on the right. You can try out the decoding program with the coded messages in the pictures below.

```
30 PRINT "TYPE THE CODE MESSAGE"

110 PRINT:PRINT "DECODED MESSAGE:"
```

When you type in a coded message make sure you type all the spaces. In these messages there is one space between each word, but there may be double spaces in other messages.

Puffing train program

The program below makes a train puff across the screen. When you type in the program, make sure you type all the letters and spaces just as they are printed here. To help you do this, the spaces are marked with the symbol □. Whenever you see this, press the space bar once.

```
10 CLS
20 LET A=0:LET B=29:LET C=1
30 FOR X=A TO B STEP C
40 PRINT TAB(0,4)
50 PRINT TAB(X);"□XX□□!□□H□"
60 PRINT TAB(X);"□□ XXXXXXX□"
70 PRINT TAB(X);"□XXXXXXXX□"
80 PRINT TAB(X);"□□00□0□00□"
90 PRINT TAB(X+7,4);"□*□"
100 IF X/2=INT(X/2) THEN
PRINT TAB(X+8,3);" * "
110 PRINT TAB(X,3);"□"
120 FOR K=1 TO 100:NEXT K
130 NEXT X
```

Remember to check the conversion pages. The instruction in line 20 is different for most computers.

The rest of this line is printed below. Don't press RETURN until you get to the end.

Check the conversion pages to see if the lines on red stripes need changing.

Running the program

When you run the program, the train appears on the left of the screen and moves to the right, puffing out smoke as it goes. You can see what it looks like in the boxes on the right.



Adding a railway track

The program lines on the right put a railway track on the screen for the train to run along. To add them to your program, type LIST. Then type the new lines.

```
23 FOR X=A TO B+10
25 PRINT TAB(X,9);"="
27 NEXT X
```

Look at the conversion pages to see if you need to change this line.



The computer draws a track like this on the screen.

Changing direction

1

```
115 IF X+16>=A+10 THEN GOTO 120
117 PRINT TAB(X+16,3);"□"
140 LET A$=INKEY$
150 IF A$="" THEN GOTO 140
160 IF A$="P" THEN GOTO 20
170 LET Z=A:LET A=B:LET B=Z:LET C=-1
180 GOTO 30
```

You can make the train move backwards as well as forwards by adding the new lines shown above.

2

Now when you run the program, you press the Q key to make the train move

3



backwards and the P key to make it move forwards.

Graphics train



On pages 38-39 you can find out how to make the computer draw a more realistic looking train using special computer graphics commands.

Computer postcard

The program below will help you write holiday postcards. You could use it to make joke postcards and send them to your friends.




```
10 CLS
20 PRINT "WHERE ARE YOU?":INPUT P$
30 PRINT "WHAT SORT OF TIME"
40 PRINT "ARE YOU HAVING?":INPUT D$
50 PRINT "DESCRIBE THE WEATHER"
60 INPUT W$
70 PRINT "AND THE FOOD":INPUT F$
80 PRINT "WHAT'S YOUR NAME?":INPUT N$
90 PRINT "WHO ARE YOU WRITING TO?"
100 INPUT A$
110 CLS
120 PRINT:PRINT "DEAR ";A$;",":PRINT
130 PRINT "I AM HAVING A ";D$
140 PRINT "TIME HERE IN ";P$;"."
150 PRINT "THE WEATHER IS ";W$
160 PRINT "AND THE FOOD IS ";F$;"."
170 PRINT "WISH YOU WERE HERE!"
180 PRINT:PRINT "LOVE FROM ";N$
```

How to use the program

When you run the program, the computer asks questions about your holiday. Type in your answers, pressing RETURN after each one. Then the computer clears the screen and writes a message using your answers.

Writing thank-you letters

Slime Mansions
Clammy Crescent
Creepton-under-Mould.
10th January 1984

Dear Aunt Mabel,
Thankyou very much
for the unusual spotty hat
that you gave me for my
birthday present.
It was a big surprise!
Love from Slimy Sid

You can turn the postcard program into a useful thank-you letter program. To change the program you type in the lines in the box below. The new program works in the same way as the postcard program. The computer makes up a message for a letter, like the one above.

If you have a cassette recorder, you can save the program on tape and use it each time you want to write a thank-you letter.

Thank-you letter lines

```
 20 PRINT "WHAT PRESENT DID YOU GET?"
 30 INPUT P$
 40
 50 PRINT "WHEN DID YOU GET IT?"
 70 PRINT "DESCRIBE IT":INPUT D$
130 PRINT "THANK YOU VERY MUCH"
140 PRINT "FOR THE ";D$;" ";P$
150 PRINT "THAT YOU GAVE ME FOR"
160 PRINT "MY ";W$;" PRESENT."
165 LET R=INT(RND(1)*5+1)
170 FOR K=1 TO R:READ M$
175 NEXT K:PRINT M$
190 DATA "IT WAS A BIG SURPRISE!"
200 DATA "IT'S JUST WHAT I WANTED."
210 DATA
220 DATA
230 DATA
```

Type the number 40 and press RETURN to get rid of line 40.

Invent some more messages and type them in quotes after the word DATA in lines 210-230.

How to use the new program



First you answer the computer's questions. Then it writes a letter on the screen which ends with one of your messages.

Birthday fact-finder

Computers are very good at calculations and the program below will make a computer calculate how old you are in years, months and days. Try using the program to find out your exact age. Then you can add extra program lines to work out your age in the year 2000 and the day of the week that you were born on.

```
10 CLS:DIM N(12)
20 FOR K=1 TO 12:READ N(K):NEXT K
30 PRINT "WHAT IS TODAY'S DATE?"
40 INPUT D1,M1,Y1
50 PRINT "WHEN WERE YOU BORN?"
60 INPUT D2,M2,Y2:LET DA=D1-D2
70 LET MA=M1-M2:LET YA=Y1-Y2
80 IF DA>=0 THEN GOTO 100
90 LET DA=DA+N(M1):LET MA=MA-1
100 IF MA>=0 THEN GOTO 120
110 LET MA=MA+12:LET YA=YA-1
120 IF YA/4<>INT(YA/4) THEN GOTO 140
130 IF M1=3 AND M2=2 THEN LET DA=DA+1
140 PRINT "YOUR AGE TODAY IS"
150 PRINT YA;" YEARS ";MA;" MONTHS"
160 PRINT DA;" DAYS"
500 DATA 31,31,28,31,30,31
510 DATA 30,31,31,30,31,30
```

This jump in the line numbers enables you to add extra lines later.

Year 2000

```
170 PRINT "  IN YEAR 2000 YOU WILL BE"
180 PRINT "  ";2000-Y2;"  YEARS OLD"
```

If you add these lines to the program, the computer will work out how old you will be in the year 2000.

Using the program

First the computer asks you today's date. You must type the date in figures as shown below.



First type the day of the month and press RETURN. Then type the month and press RETURN and then the year and press RETURN.



When the computer asks, type your date of birth in the same way. The computer will do a quick calculation and tell you exactly how old you are.

Day-finder

To make the computer work out what day of the week you were born on you need to add all the lines below to your program. There are some complicated calculations in lines 400 to 440 so make sure you type them carefully.




```
 15 DIM D$(8)
 25 FOR K=1 TO 8:READ D$(K):NEXT K
190 LET D=D2:LET M=M2:LET Y=Y2
200 GOSUB 400
210 PRINT "YOU WERE BORN"
220 PRINT "ON A ";D$(X)
230 STOP
400 LET X=Y-1901
410 LET E=INT(X/4)+(X*365)+D-30
420 FOR K=1 TO M:LET E=E+N(K):NEXT K
430 LET X=INT(7*((E/7)-INT(E/7))+.5)+1
440 IF M>2 AND Y/4=INT(Y/4) THEN LET X=X+1
450 RETURN
520 DATA "SUNDAY","MONDAY","TUESDAY","WEDNESDAY"
530 DATA "THURSDAY","FRIDAY","SATURDAY","SUNDAY"
```

On the Spectrum, press the key marked **RETURN** here.

On page 35 there are some more lines to add to this program. They will make the computer give you all sorts of information about your next birthday.

Star racer

This is a program to make stars race each other across your computer screen. Opposite you can find out how to turn the race into a game you can play with a friend. Before typing the program lines check whether you need to make any conversions for your computer.



You should type a space before the star inside quotes in this line.

```
10 CLS
20 DIM X(6):LET Y=0
30 FOR K=1 TO 6
40 PRINT TAB(1,K+K);K
50 LET X(K)=2:NEXT K
60 FOR K=1 TO 6
70 PRINT TAB(X(K),K+K);" *"
80 IF X(K)=37 AND Y=0 THEN LET Y=K
90 LET R=INT(RND(1)*2+1)
100 IF R=1 THEN LET X(K)=X(K)+1
110 NEXT K
120 IF Y=0 THEN GOTO 60
130 PRINT:PRINT "NUMBER ";Y;" WINS"
```

Running the race



When you run the program, the computer puts six stars on the left side of the screen. These are the "runners" in the race and they are numbered 1 to 6.



The stars race each other across the screen and the first to reach the right wins. Every time you run the race the computer changes the speed of each star.



When the race is over, run the program a few more times and you will see that there is usually a different winner each time.

Adding a race track

```
51 PRINT TAB(0,0)
52 FOR I=1 TO 7
53 FOR L=1 TO 38
54 PRINT "-" ;: NEXT L
55 PRINT:PRINT:NEXT I
```

If you add these lines to the program, the computer will make a race track for the stars.



Now when you run the program, the computer starts by putting the race track on the screen and the stars move across the screen inside their own lanes.

Race game

You can turn the race into a game you can play with a friend by adding the lines below to the program. The new program lets you both guess the winner before the race starts. You can see how it works in the boxes on the right.

You only need to change this line if you have an Apple or Commodore computer.

```
 22 PRINT "WHAT ARE YOUR NAMES?"
 24 PRINT "PLAYER 1":INPUT A$
 26 PRINT "PLAYER 2":INPUT B$
 28 CLS:LET Y=0
 56 PRINT A$;" 'S GUESS";: INPUT G1
 58 PRINT B$;" 'S GUESS";: INPUT G2
140 PRINT:PRINT
150 IF Y=G1 THEN PRINT "WELL DONE ";A$
160 IF Y=G2 THEN PRINT "WELL DONE ";B$
170 INPUT R$:GOTO 28
```

If you have a Spectrum you must add two extra lines so look at the conversion pages.

Playing the game

First the computer asks each of you your name. Type it in and press RETURN.

Then it asks for your guesses. Choose a star and type in its number. If your star wins, the computer sends you a message after the race. To play again, press RETURN.

Obstacle course game

In this game you have to pit your wits against the computer as you steer a star round obstacles on the screen. Try it and see if you can complete the course. Then you can add some program lines to make the game more difficult.

On some computers you need to add extra lines to this program so check the conversion pages.

```
10 LET A=38:LET B=23:LET Z$=CHR$(255)
20 CLS:DIM Z(A,B)
30 LET Q=1:LET W=15
40 PRINT TAB(Q,W);" * "
50 IF Z(Q,W)=1 THEN PRINT "CRASH":STOP
60 IF A=Q THEN PRINT "WELL DONE":STOP
70 LET A$=INKEY$
80 LET X=INT(RND(1)*A+1)
90 LET Y=INT(RND(1)*B+1)
100 LET Z(X,Y)=1:PRINT TAB(X,Y);Z$
110 IF A$="A" THEN LET W=W-1
120 IF A$="Z" THEN LET W=W+1
130 IF A$="M" THEN LET Q=Q+1
140 GOTO 40
```

If you crash into an obstacle, the program stops and you lose the game.

How to play



As soon as you type RUN a star appears on the left of your screen as shown in the picture above.



Then the screen starts filling up with small squares. You must move the star across the screen avoiding all these "obstacles".*



To move the star, there are 3 control keys. Press the A key to move the star up, Z to move it down and M to move it across to the right.

*On the Apple, the obstacles are 0's instead of squares.

Improving the game



These pictures show what the game looks like when you add the lines below to the program. First the computer draws a "wall" round the edge of the screen with a gap in each side.

The star is in the gap on the left and the gap on the right is the goal. The computer starts filling the area inside the walls with the obstacle blocks.



To win the game you must steer the star through the blocks and into the goal on the right. If you bump into an obstacle or the wall, you lose the game.

Adding the extra lines

To play the new version of this game, type in the lines on the right. Check the conversion pages to see whether you need to change lines 32 and 34 to suit your computer.

```
22 FOR K=1 TO A:LET Z(K,1)=1
24 LET Z(K,B)=1:NEXT K
26 FOR L=1 TO B:LET Z(1,L)=1
28 LET Z(A,L)=1:NEXT L
31 LET Z(Q,W)=Ø
32 LET Z(A,INT(RND(1)*15+3))=2
33 FOR L=1 TO B:FOR K=1 TO A
34 IF Z(K,L)=1 THEN PRINT TAB(K,L);Z$
35 NEXT K:NEXT L
6Ø IF Z(Q,W)=2 THEN PRINT "WELL DONE":STOP
```



On page 37 you can find out how to add sound effects to this game.

Find the key

How fast are your reactions? The program opposite is for a game which tests your speed at finding keys on a computer keyboard. You can find out how to play the game below and then there is an idea for improving it.



How to play the game

When you run the program, the computer tells you to find a key.

When you find the key, press it and the computer will tell you how long you took to find it.

Then check your time. Get ready for the next go and press RETURN to play again.

Find the key program

```
10 CLS:LET N=0:LET A=ASC("A")
20 LET X=INT(RND(1)*26+A)
30 PRINT:PRINT "FIND ";CHR$(X)
40 LET A$=INKEY$
50 IF A$="" THEN LET N=N+1:GOTO 40
60 IF A$<>CHR$(X) THEN GOTO 120
70 LET S=INT((N/235)*10)/10
```

Line 70 is different for most computers so make sure you check the conversion pages.

```
80 PRINT "YOU TOOK ";S;" SECONDS"
90 PRINT "PRESS RETURN TO CONTINUE"
100 INPUT R$
110 GOTO 10
120 PRINT "WRONG KEY":GOTO 90
```

If you have a Spectrum type ENTER instead of RETURN.

Improving the game

To make the game better, you can add the lines on the right. Now when you want to finish the game, type STOP and press RETURN. When you do this, the computer will tell you your best time, worst time and your average speed for finding the keys.

```
 5 LET B=100:LET W=0
 7 LET T=0:LET K=0
74 IF S<B THEN LET B=S
75 IF S>W THEN LET W=S
76 LET T=T+S:LET K=K+1
105 IF R$="STOP" THEN GOTO 130
130 LET T=T/K:LET A=(INT(T*10))/10
140 PRINT "BEST TIME :";B;" SECONDS"
150 PRINT "WORST TIME:";W;" SECONDS"
160 PRINT "AVERAGE :";A;" SECONDS"
```

Number puzzle programs

The programs on these two pages are for number games. In Guess the number the computer chooses a number and you have to try and guess it. In Mindreader you choose a number and the computer works out what it is.



Guess the number

```
10 CLS
20 LET N=0
30 LET M=100
40 PRINT:PRINT
50 LET R=INT(RND(1)*M+1)
60 PRINT "THINK OF A NUMBER"
70 PRINT "BETWEEN 1 AND ";M
80 LET N=N+1
90 PRINT "WHAT'S YOUR GUESS?"
100 INPUT G
110 IF G=R THEN GOTO 180
120 IF G<R THEN PRINT "TRY A BIGGER NUMBER"
130 IF G>R THEN PRINT "TRY A SMALLER NUMBER"
140 IF N<10 THEN GOTO 80
150 PRINT "YOU LOSE"
160 PRINT "THE NUMBER WAS ";R
170 STOP
180 PRINT "WELL DONE"
190 PRINT "YOU GOT IT IN ";N
```

When you run the program, the computer chooses a number between 1 and 100. You have ten chances to guess what it is. If your guess is wrong, the computer tells you whether to pick a bigger or a smaller number next time.

How to make the game easier

1

```
140 IF N<15 THEN GOTO 80
```

If you find the game too difficult, change line 140 as shown above. This gives you 15 chances to guess the computer's number.

2

```
30 LET M=50
```

This is the number to change. It tells the computer the highest number it can choose.

Another way to make the game easier is to make the computer choose numbers between 1 and 50, or 1 and 20. You can do this by changing the number in line 30 as shown above.

Mindreader



In Mindreader the computer tells you to think of a number between 1 and 100. Do not type it in, just press RETURN when you have thought of one. To work out the number, the computer asks you for the answers to several sums, so have a pen and paper handy.

Mindreader program

```
 10 CLS
 20 PRINT "THINK OF A NUMBER"
 30 PRINT "BETWEEN 1 and 100"
 40 PRINT "THEN PRESS RETURN"
 50 INPUT R$:PRINT
 60 PRINT "DIVIDE THE NUMBER BY 3"
 70 PRINT "WHAT IS THE REMAINDER?"
 80 INPUT A
 90 IF A>=0 AND A<=2 THEN GOTO 110
100 GOSUB 280:GOTO 60
110 PRINT "DIVIDE THE NUMBER BY 5"
120 PRINT "WHAT IS THE REMAINDER?"
130 INPUT B
140 IF B>=0 AND B<=4 THEN GOTO 160
150 GOSUB 280:GOTO 110
160 PRINT "DIVIDE THE NUMBER BY 7"
170 PRINT "WHAT IS THE REMAINDER?"
180 INPUT C
190 IF C>=0 AND C<=6 THEN GOTO 210
200 GOSUB 280:GOTO 160
210 LET D=70*A+21*B+15*C
220 IF D<=105 THEN GOTO 240
230 LET D=D-105:GOTO 220
240 CLS
250 PRINT:PRINT:PRINT
260 PRINT "YOUR NUMBER IS ";D
270 STOP
280 PRINT "THAT'S NOT RIGHT"
290 PRINT "TRY AGAIN"
300 RETURN
```

Geography quiz

You can test your knowledge of the world with the geography quiz program on the right. The computer asks you for the names of the capital cities of 16 different countries. At the end of the quiz the computer gives you a score. Try running the program and see how many answers you get right.

```
10 CLS
20 LET S=0
30 DIM A$(16),B$(16)
40 FOR K=1 TO 16:READ A$(K):NEXT K
50 FOR K=1 TO 16:READ B$(K):NEXT K
60 FOR L=1 TO 16
70 PRINT:PRINT:PRINT
80 PRINT "WHAT IS THE CAPITAL"
90 PRINT "CITY OF ";B$(L);"?"
100 INPUT R$
110 IF R$=A$(L) THEN GOTO 150
120 PRINT "NO. THE CAPITAL OF"
130 PRINT B$(L);" IS ";A$(L)
140 GOTO 160
150 PRINT "WELL DONE":LET S=S+1
160 FOR K=1 TO 1000:NEXT K
170 NEXT L
180 CLS
190 PRINT:PRINT:PRINT
200 PRINT "YOU GOT ";S;" CORRECT"
210 DATA "COPENHAGEN", "PARIS", "ROME"
220 DATA "BRUSSELS", "PEKING", "BONN"
230 DATA "STOCKHOLM", "MOSCOW", "CANBERRA"
240 DATA "OSLO", "LISBON", "VIENNA", "CAIRO"
250 DATA "LONDON", "ATHENS", "MADRID"
260 DATA "DENMARK", "FRANCE", "ITALY"
270 DATA "BELGIUM", "CHINA", "WEST GERMANY"
280 DATA "SWEDEN", "RUSSIA", "AUSTRALIA"
290 DATA "NORWAY", "PORTUGAL", "AUSTRIA"
300 DATA "EGYPT", "ENGLAND", "GREECE", "SPAIN"
```

If you have a Spectrum you will need to add an extra line, so look at the conversion pages.

Times table quiz

This program helps you with your maths. It makes the computer test your times tables by setting you ten sums. Each time you run the program, the sums are different. Run the program a few times and see what scores you get.

```
10 LET N=0
20 FOR K=1 TO 10
30 CLS
40 PRINT:PRINT:PRINT
50 LET X=INT(RND(1)*12+1)
60 LET Y=INT(RND(1)*12+1)
70 PRINT "WHAT IS ";X;" X ";Y
80 INPUT A
90 IF A=X*Y THEN GOTO 110
100 PRINT "WRONG. IT'S ";X*Y:GOTO 120
110 PRINT "CORRECT":LET N=N+1
120 PRINT:PRINT
130 PRINT "PRESS RETURN"
140 INPUT A$
150 NEXT K
160 PRINT "YOU SCORED ";N
```

More birthday facts

```
230 LET Y=Y1
240 IF M2<M1 THEN LET Y=Y+1
250 IF M2=M1 AND D2<D1 THEN LET Y=Y+1
260 GOSUB 400
270 PRINT "YOUR NEXT BIRTHDAY IS ON A ";D$(X)
280 LET Q1=D1-31:LET Q2=D2-31
290 FOR K=1 TO M1:LET Q1=Q1+N(K):NEXT K
300 FOR K=1 TO M2:LET Q2=Q2+N(K):NEXT K
310 LET Q=Q1-Q2
320 IF Q>0 THEN LET Q=365-Q
330 IF Q<0 THEN LET Q=ABS(Q)
340 PRINT "IT IS IN"
350 PRINT Q;" DAYS TIME AND"
360 PRINT "YOU WILL BE ";YA+1;" YEARS OLD"
370 STOP
```



Here are some extra lines to add to the Birthday fact-finder program on page 24. They make the computer tell you all sorts of things about your next birthday. Before you type them in, you need to type in all the program lines on pages 24 and 25.

Adding sound and colour

On these two pages there are extra program lines which will add sound effects to the Obstacle course game on page 28 and colour to the Drawing pictures program on page 14. In each case type in the whole program before adding the extra lines for your computer.

Drawing pictures in colour

When you have added the lines, you can change the colour of the star by pressing C and the background colour by pressing P or B (try both).

Commodore 64

```
 25 LET C=14:LET B=14:LET P=6
 41 IF A$="P" THEN LET P=P+1
 42 IF A$="B" THEN LET B=B+1
 43 IF A$="C" THEN LET C=C+1
 44 IF P>15 THEN LET P=0
 45 IF B>15  THEN LET B=0
 46 IF C>15 THEN LET C=0
 47 POKE 53281,P
 48 POKE 53280,B
 95 POKE 55336+X+Y*40,C
```

Apple*




```
 10 GR
 25 LET C=17
 45 IF A$="C" THEN LET C=C+17
 46 IF C>255 THEN LET C=0
 90 COLOR=C:PLOT X,Y
110 COLOR=0:PLOT X,Y
```

VIC 20

```
  9 INPUT S$:LET S=PEEK(7748)
 25 LET B=4:LET P=1:LET C=6
 26 POKE 8400+X+Y*22,C
 27 POKE 7680+X+Y*22,S
 41 IF A$="P" THEN LET P=P+1
 42 IF A$="B" THEN LET B=B+1
 43 IF A$="C" THEN LET C=C+1
 44 IF P>15 THEN LET P=0
 45 IF B>7 THEN LET B=0
 46 IF C>7 THEN LET C=0
 47 POKE 36879,(8+16*P+B)
 82 PRINT CHR$(19);
 90 POKE 36878,15:POKE 36875,225
 92 FOR T=1 TO 100:NEXT T
 95 POKE 36875,0
100 GOTO 26
```

Spectrum

```
  1 LET B=6
  2 PAPER B
 25 LET C=0
 41 IF A$="P" OR A$="B" THEN LET B=B+1
 43 IF A$="C" THEN LET C=C+1
 45 IF B>7 THEN LET B=0
 46 IF C>7 THEN LET C=0
 47 BORDER B
 90 PRINT AT Y,X;INK C;S$
```

BBC and Electron

```
  3 MODE 2
 25 LET C=7:LET B=128
 26 COLOUR B
 42 IF A$="C" THEN LET C=C+1
 43 IF C>7 THEN LET C=0
 45 COLOUR C
 46 IF A$="B" OR A$="P" THEN LET B=B+1
 47 IF B>135 THEN LET B=128
 48 VDU 19,128,B;0;
```


*On the Apple you cannot change the background colour and the pictures are now drawn with coloured squares.

Sound effects for Obstacle course game

When you add the extra lines to the Obstacle course game, the computer produces a sound effect when you win the game and a different one if you crash.*

Commodore 64

```
 50 IF Z(Q,W)=1 THEN GOSUB 400:END
 60 IF Z(Q,W)=2 THEN GOSUB 200:END
200 PRINT CHR$(19)
210 FOR J=1 TO 20:PRINT:NEXT J
220 PRINT TAB(15);"WELL DONE"
230 POKE 54296,15:POKE 54276,17
240 POKE 54277,129:POKE 54278,64
250 FOR H=0 TO 200:POKE 54273,H:NEXT H
260 POKE 54272,0:POKE 54273,0
270 RETURN
400 PRINT CHR$(19)
410 FOR J=1 TO 20:PRINT:NEXT J
420 PRINT TAB(15);"CRRRAAASH!"
430 FOR D=20 TO 0 STEP -1:POKE 54296,D
440 POKE 54276,129:POKE 54277,15
450 POKE 54273,40:POKE 54272,200:NEXT D
460 POKE 54276,0:POKE 54277,0:POKE 54273,0
470 RETURN
```

Apple

```
 50 IF Z(Q,W)=1 THEN GOSUB 400:END
 60 IF Z(Q,W)=2 THEN GOSUB 200:END
200 PRINT "WELL DONE!"
210 FOR N=1 TO 5:FOR T=1 TO 3
220 PRINT CHR$(135)
230 FOR P=1 TO 10:NEXT P
240 NEXT T:FOR P=1 TO 50:NEXT P
250 NEXT N:RETURN
400 PRINT "CRRRAAASH!"
410 FOR N=1 TO 100:SOUND=PEEK(-16336)
420 NEXT N:RETURN
```

VIC 20

```
 50 IF Z(Q,W)=1 THEN GOSUB 400:END
 60 IF Z(Q,W)=2 THEN GOSUB 200:END
200 PRINT CHR$(19)
210 FOR J=1 TO 20:PRINT:NEXT J
220 PRINT TAB(5);"WELL DONE"
230 FOR E=1 TO 25:NEXT E
240 GOSUB 500:RETURN
400 PRINT CHR$(19)
410 FOR J=1 TO 20:PRINT:NEXT J
420 PRINT TAB(5);"CRRRAAAASH!"
430 POKE 36877,200
440 FOR D=15 TO 0 STEP -1:POKE 36878,D
450 FOR T=1 TO 40:NEXT T:NEXT D
460 POKE 36877,0:POKE 36878,0
470 RETURN
500 POKE 36878,15:FOR D=130 TO 240
520 POKE 36876,D:FOR T=1 TO 5:NEXT T
530 NEXT D:POKE 36876,0:POKE 36878,0
540 RETURN
```

Spectrum

```
 50 IF Z(Q,W)=1 THEN GOSUB 400:STOP
 60 IF Z(Q,W)=2 THEN GOSUB 200:STOP
200 FOR N=0 TO 50 STEP 2
210 BEEP .1,N:BEEP .1,N-2:NEXT N
220 PRINT AT 10,9;"WELL DONE"
230 RETURN
400 PRINT AT 10,9;"CRRRAAASH!"
410 FOR N=50 TO 0 STEP -1
420 BEEP .1,N:NEXT N
430 RETURN
```

BBC and Electron

```
 50 IF Z(Q,W)=1 THEN GOSUB 400:END
 60 IF Z(Q,W)=2 THEN GOSUB 200:END
200 PRINT TAB(5,27)"WELL DONE"
210 FOR N=0 TO 120:SOUND 1,-15,N,1
220 NEXT N:RETURN
400 PRINT TAB(7,27)"CRRRAAASH!"
410 FOR N=0 TO 120:SOUND 1,-15,120-N,1
420 NEXT N:CLS:RETURN
```

*To hear the sound effects on the Commodore 64 and VIC 20 you may need to adjust the TV's volume control.

Puffing train graphics

Below and on the next page are lines to add to the Puffing train program on page 20. They make the computer draw a coloured train, and as it puffs along you can make it whistle by pressing the W key.

The lines are different for each computer, so make sure you type in the right ones for yours. Remember, too, that you must type in all the program lines on pages 20-21 before adding the extra lines.



BBC and Electron

```
 2 MODE 1:COLOUR 130
 4 VDU 23,224,255,255,255,255,255,255,255,255
 6 VDU 23,225,225,255,255,255,0,0,0,0
 8 VDU 23,226,0,0,0,0,255,255,255,255
12 LET T$=CHR$(224)+CHR$(224)+CHR$(224)+CHR$(224)
   +CHR$(224)+CHR$(224)+CHR$(224)+CHR$(32)
14 LET T1$=CHR$(32)+CHR$(224)+CHR$(224)
   +CHR$(32)+CHR$(32)+CHR$(226)+CHR$(32)
   +CHR$(32)+CHR$(224)+CHR$(32)
16 LET T2$=CHR$(32)+CHR$(32)+T$
18 LET T3$=CHR$(32)+CHR$(224)+T$
22 COLOUR 0
25 PRINT TAB(X,9)CHR$(225)
45 COLOUR 1
50 PRINT TAB(X)T1$
60 PRINT TAB(X)T2$
70 PRINT TAB(X)T3$
75 COLOUR 0
120 SOUND 1,-9.0,1
122 SOUND 1,-9,5,1
123 LET A$=INKEY$(0)
125 IF A$="W" THEN VDU 7
```

Apple

The lines below make the Apple produce the train's whistle. There are no graphics instructions because they are very complicated and would take a long time to type in.

```
114 GOSUB 500
120 A$="":IF PEEK(-16384)>127 THEN GET A$
125 IF A$<>"W" THEN GOTO 130
126 PRINT CHR$(135):FOR N=1 TO 100
127 NEXT N:PRINT CHR$(135)
500 FOR T=1 TO 10:SOUND=PEEK(-16336)
510 FOR Q=1 TO RND(1)*15:NEXT Q
520 NEXT T:RETURN
```

Commodore 64

```
  5 POKE 54296,15:POKE 54283,17
  7 POKE 54284,129:POKE 54285,128
 12 POKE 53280,14:POKE 53281,14
 25 PRINT TAB(X);CHR$(144);CHR$(184)
 46 LET R$=CHR$(18):LET O$=CHR$(146):LET C$=CHR$(28)
 50 PRINT TAB(X);C$;"□";R$;"□□";O$;"□□";
 55 PRINT CHR$(185);"□□";CHR$(161);"□"
 60 PRINT TAB(X);C$;"□□";R$;"□□□□□□□";O$;"□"
 70 PRINT TAB(X);C$;"□";R$;"□□□□□□□□";O$;"□"
 80 PRINT TAB(X);CHR$(144);"□";O$;"□OO□O□OO□"
 90 PRINT "↑↑↑↑↑";TAB(X+7);CHR$(5);"□*□"
100 IF X/2=INT(X/2) THEN PRINT "↑↑";TAB(X+8);CHR$(5);"* "
120 GET A$
123 IF A$<>"W" THEN GOTO 130
124 FOR R=0 TO 1:POKE 54296,15
125 POKE 54276,33:POKE 54277,129
126 FOR T=1 TO 25:POKE 54273,68:POKE 54272,149:NEXT T
127 POKE 54273,0:POKE 54272,0:POKE 54276,0:NEXT R
130 POKE 54296,15:POKE 54283,17:POKE 54284,129
132 FOR T=1 TO 15:POKE 54280,17:POKE 54279,37:NEXT T
134 POKE 54280,0:POKE 54279,0:POKE 54283,0
135 NEXT X
```

This symbol means type a space.

This symbol means you should press the cursor up/down key with SHIFT.

VIC 20

```
  5 POKE 36878,15
 12 POKE 36879,59
```

For lines 25 to 123 type the lines listed for the Commodore 64.

```
124 FOR R=0 TO 1:POKE 36878,15
125 POKE 36875,225:FOR T=1 TO 200:NEXT T
126 POKE 36875,0:NEXT R
130 POKE 36878,15:POKE 36874,175
131 POKE 36877,175:FOR T=1 TO 25:NEXT T
132 POKE 36874,195:POKE 36877,195
133 FOR T=1 TO 25:NEXT T
134 POKE 36878,0:POKE 36874,0:POKE 36877,0
135 NEXT X
```

Spectrum

```
  5 BORDER 5:PAPER 6
 25 PRINT AT 9,X;INK 1;"▬"
 45 INK 2
```

These graphics characters are on the number keys. Find the key marked with the shape you want and press it with SHIFT.

```
 50 PRINT TAB(X);"□■■□□▄□□■□"
 60 PRINT TAB(X);"□□■■■■■■□"
 70 PRINT TAB(X);"□■■■■■■■□"
 75 INK 0
120 BEEP .1,-20:BEEP.1,-15
123 LET A$=INKEY$
125 IF A$="W" THEN BEEP .5,20:BEEP .5,20
```

Program conversions: BBC and Commodore 64

Here and on the next few pages are lines you may need to type instead of those on red (or grey) stripes. To find a line, look in the section for your computer and find the name and page number of the program you are typing in. Then look for a line with the same number as the one on the stripe. If you cannot find one, you can type the line on the stripe just as it is.

BBC conversions

Page 14
Drawing pictures
30 LET A$=INKEY$(0)

Page 16
Computer Snap
60 LET A$=INKEY$(0)

Page 21
Puffing train program
Changing direction
140 LET A$=INKEY$(0)

Page 28
Obstacle course game
70 LET A$=INKEY$(0)

Page 31
Find the key
40 LET A$=INKEY$(0)
70 LET S=INT((N/316)*10)/10

Commodore 64 conversions

Page 7
Special words
10 PRINT CHR$(147)

Page 8
Happy Birthday
10 PRINT CHR$(147)

Page 10
Space monsters
10 PRINT CHR$(147)

Page 11
Making the monster move
120 PRINT CHR$(147)

Page 12
Red Alert
10 PRINT CHR$(147)

Page 13
Personal messages
2 PRINT CHR$(147)



Page 14
Drawing pictures
10 PRINT CHR$(147)
30 GET A$
82 PRINT CHR$(19)
84 FOR K=1 TO Y:PRINT:NEXT K
90 PRINT TAB(X);" * "

Page 15
Rubbing out stars
110 PRINT TAB(X);" "

Patterns with different symbols
3 PRINT CHR$(147)
90 PRINT TAB(X);S$

Page 16 Computer Snap
10 PRINT CHR$(147)
60 GET A$

Each time you see this symbol, you should press this key with SHIFT.

Page 18
Code-writer
```
 10 PRINT CHR$(147)
100 PRINT CHR$(147)
```

Page 20
Puffing train program
```
 10 PRINT CHR$(147)
 40 PRINT CHR$(19)
 45 PRINT:PRINT:PRINT:PRINT
 90 PRINT "↑↑↑↑↑";TAB(X+7);"□*□"
100 IF X/2=INT(X/2) THEN PRINT "↑↑";TAB(X+8);"*"
110 PRINT "↑";TAB(X);"□"
```

Page 21
Adding a railway track
```
24 PRINT CHR$(19):FOR Y=1 TO 8:PRINT:NEXT Y
25 PRINT TAB(X);"="
```

Add line 24 to the program.

Changing direction
```
117 PRINT "↑";TAB(X+16);"□"
140 GET A$
```

Page 22
Computer postcard
```
 10 PRINT CHR$(147)
110 PRINT CHR$(147)
```

Page 24
Birthday fact-finder
```
10 PRINT CHR$(147):DIM N(12)
```

Add line 59 to the program.

Page 26
Star racer
```
 10 PRINT CHR$(147)
 40 PRINT:PRINT TAB(0);K
 59 PRINT CHR$(19)
 70 PRINT:PRINT TAB(X(K));"  *"
120 IF Y=0 THEN GOTO 59
```

Page 27
Adding a race track
```
51 PRINT CHR$(19)
```

Race game
```
28 PRINT CHR$(147):LET Y=0
```

Page 28
Obstacle course game
```
 10 LET A=38:LET B=20
 15 LET Z$=CHR$(18)+CHR$(160)+CHR$(146)
 20 PRINT CHR$(147):DIM Z(A,B)
 40 PRINT CHR$(19)
 43 FOR K=1 TO W:PRINT:NEXT K
 45 PRINT TAB(Q);" * "
 70 GET A$
100 LET Z(X,Y)=1:PRINT CHR$(19)
103 FOR K=1 TO Y:PRINT:NEXT K
105 PRINT TAB(X);Z$
```

There are five extra lines here. Make sure you add them to the program.

Page 29
Improving the game
```
 34 IF Z(K,L)=1 THEN GOSUB 150
150 PRINT CHR$(19)
160 FOR J=1 TO L:PRINT:NEXT J
170 PRINT TAB(K);Z$
180 RETURN
```

Add these extra lines to the program.

Page 31
Find the key
```
10 PRINT CHR$(147):LET N=0:LET A=ASC("A")
40 GET A$
70 LET S=INT((N/115)*10)/10
```

Page 32
Guess the number
```
10 PRINT CHR$(147)
```

Page 33
Mindreader
```
 10 PRINT CHR$(147)
240 PRINT CHR$(147)
```

Page 34
Geography quiz
```
 10 PRINT CHR$(147)
180 PRINT CHR$(147)
```

Page 35
Times table quiz
```
30 PRINT CHR$(147)
```

Program conversions: Electron and VIC 20

Electron conversions

Page 14
Drawing pictures
```
30 LET A$= INKEY$(0)
```

Page 16
Computer Snap
```
60 LET A$=INKEY$(0)
```

Page 21
Puffing train program
Changing direction
```
140 LET A$=INKEY$(0)
```

Page 28
Obstacle course game
```
  5 VDU 23,255,255,255,255,255,255,255,255,255
70 LET A$=INKEY$(0)
```

Page 31
Find the key
```
40 LET A$=INKEY$(0)
```

VIC 20 conversions

Page 7
Special words
```
10 PRINT CHR$(147)
```

Page 8
Happy Birthday
```
10 PRINT CHR$(147)
```

Page 10
Space monsters
```
10 PRINT CHR$(147)
```

Page 11
Making the monster move
```
120 PRINT CHR$(147)
```

Page 12
Red Alert
```
10 PRINT CHR$(147)
```

Page 13
Personal messages
```
2 PRINT CHR$(147)
```



Page 14
Drawing pictures
```
10 PRINT CHR$(147)
30 GET A$
82 PRINT CHR$(19)
84 FOR K=1 TO Y:PRINT:NEXT K
90 PRINT TAB(X);"*"
```

Page 15
Rubbing out stars
```
110 PRINT TAB(X);" "
```

Patterns with different symbols
```
  3 PRINT CHR$(147)
90 PRINT TAB(X);S$
```

Page 16
Computer Snap
```
10 PRINT CHR$(147)
60 GET A$
```

Page 18
Code-writer
```
  10 PRINT CHR$(147)
100 PRINT CHR$(147)
```



Page 20
Puffing train program
```
10 PRINT CHR$(147)
20 LET A=0:LET B=10:LET C=1
40 PRINT CHR$(19)
45 PRINT:PRINT:PRINT:PRINT
90 PRINT "↑↑↑↑↑";TAB(X+7);"□*□"
100 IF X/2=INT(X/2) THEN PRINT "↑↑";TAB(X+8);"*"
110 PRINT "↑";TAB(X);"□"
```

Page 21

Adding a railway track

```
24 PRINT CHR$(19):FOR Y=1 TO 8:PRINT:NEXT Y
25 PRINT TAB(X);"="
```

Changing direction

```
117 PRINT "↑";TAB(X+16);"□"
140 GET A$
```

Page 22

Computer postcard

```
 10 PRINT CHR$(147)
110 PRINT CHR$(147)
```

Page 24

Birthday fact-finder

```
10 PRINT CHR$(147):DIM N(12)
```

Page 26

Star racer




```
 10 PRINT CHR$(147)
 40 PRINT:PRINT TAB(0);K
 59 PRINT CHR$(19)
 70 PRINT:PRINT TAB(X(K));"  *"
 80 IF X(K)=19 AND Y=0 THEN LET Y=K
120 IF Y=0 THEN GOTO 59
```

Page 27

Adding a race track

```
51 PRINT CHR$(19)
53 FOR L=1 TO 21
```

Race game

```
28 PRINT CHR$(147):LET Y=0
```

Page 28

Obstacle course game

```
 10 LET A=20:LET B=20
 15 LET Z$=CHR$(18)+CHR$(160)+CHR$(146)
 20 PRINT CHR$(147):DIM Z(A,B)
 40 PRINT CHR$(19)
 43 FOR K=1 TO W:PRINT:NEXT K
 45 PRINT TAB(Q);"*"
 70 GET A$
100 LET Z(X,Y)=1:PRINT CHR$(19)
103 FOR K=1 TO Y:PRINT:NEXT K
105 PRINT TAB(X);Z$
```



Page 29

Improving the game

```
 34 IF Z(K,L)=1 THEN GOSUB 150
150 PRINT CHR$(19)
160 FOR J=1 TO L:PRINT:NEXT J
170 PRINT TAB(K);Z$
180 RETURN
```



Page 31

Find the key

```
10 PRINT CHR$(147):LET N=0:LET A=ASC("A")
40 GET A$
70 LET S=INT((N/140)*10)/10
```

Page 32

Guess the number

```
10 PRINT CHR$(147)
```

Page 33

Mindreader

```
 10 PRINT CHR$(147)
240 PRINT CHR$(147)
```

Page 34

Geography quiz

```
 10 PRINT CHR$(147)
180 PRINT CHR$(147)
```

Page 35

Times table quiz

```
30 PRINT CHR$(147)
```

Program conversions: Spectrum

Page 8 Happy Birthday

```
50 PRINT A$(K TO K);
```



Page 14 Drawing pictures

```
35 FOR K=1 TO 15:NEXT K
90 PRINT AT Y, X;"*"
```

Page 15 Rubbing out stars

```
110 PRINT AT Y,X;" "
```

Patterns with different symbols

```
90 PRINT AT Y,X;S$
```

Page 16 Computer Snap

```
30 LET X=INT(RND * 10+1)
40 LET Y=INT(RND * 10+1)
```

Page 17 Word and number snap

```
2 DIM N$(10,5)
```

Page 18 Code-writer

```
70 LET C$=C$+M$(K+1 TO K+1)
80 LET C$=C$+M$(K TO K)
```

Page 20 Puffing train program

```
 20 LET A=0:LET B=20:LET C=1
 40 PRINT AT 4,0
 90 PRINT AT 4,X+7;"□ * □"
100 IF X/2=INT(X/2) THEN PRINT AT 3,X+8;"*"
110 PRINT AT 3,X;"□"
```

Page 21 Adding a railway track

```
25 PRINT AT 9,X;"="
```

Changing direction

```
117 PRINT AT 3,X+16;"□"
```

Page 23 Writing thank-you letters

```
165 LET R=INT(RND * 5+1)
```

Page 25 Day-finder

```
15 DIM D$(8,9)
```

Page 26 Star racer

```
40 PRINT AT K+K,1;K
70 PRINT AT K+K,X(K);"     *"
80 IF X(K)=30 AND Y=0 THEN LET Y=K
90 LET R=INT(RND * 2+1)
```

Page 27 Adding a race track

```
51 PRINT AT 0,0
53 FOR L=1 TO 31
```



Race game

```
57 PRINT G1
59 PRINT G2
```

Page 28 Obstacle course game

```
 10 LET A=30:LET B=20:LET Z$=CHR$(143)
 40 PRINT AT W,Q;"*"
 80 LET X=INT(RND * A+1)
 90 LET Y=INT(RND * B+1)
100 LET Z(X,Y)=1:PRINT AT Y,X;Z$
```

Page 29 Improving the game

```
32 LET Z(A,INT(RND * 15+3))=2
34 IF Z(K,L)=1 THEN PRINT AT L,K;Z$
```

Page 31 Find the key

```
 10 CLS:LET N=0:LET A=CODE("A")
 20 LET X=INT(RND * 26+A)
 70 LET S=INT((N/77) * 10)/10
103 PAUSE 20
```

Page 32 Guess the number

```
50 LET R=INT(RND*M+1)
```



Page 34 Geography quiz

```
 30 DIM A$(17,10):DIM B$(17,15)
105 IF LEN(R$)<10 THEN LET R$=R$+
    " ":GOTO 105
```

Page 35 Times table quiz

```
50 LET X=INT(RND * 12+1)
60 LET Y=INT(RND * 12+1)
```

Program conversions: Apple

Page 7
Special words
```
10 HOME
```

Page 8
Happy birthday
```
10 HOME
```

Page 10
Space monsters
```
10 HOME
```

Page 11
Making the monster move
```
120 HOME
```

Page 12
Red Alert
```
10 HOME
```

Page 13
Personal messages
```
2 HOME
```

Page 14
Drawing pictures
```
10 HOME
30 A$="":IF PEEK (-16384)>127 THEN GET A$
90 VTAB(Y):HTAB(X):PRINT "*"
```

Page 15
Rubbing out stars
```
110 VTAB(Y):HTAB(X):PRINT " "
```

Patterns with different symbols
```
 3 HOME
90 VTAB(Y):HTAB(X):PRINT S$
```

Page 16
Computer Snap
```
10 HOME
60 A$="":IF PEEK(-16384)>127 THEN GET A$
```

Page 18
Code-writer
```
 10 HOME
100 HOME
```

Page 20
Puffing train program
```
 10 HOME
 20 LET A=1:LET B=29:LET C=1
 40 VTAB(5):HTAB(1)
 90 VTAB(4):HTAB(X+7):PRINT "□*□"
100 IF X/2=INT(X/2) THEN VTAB(3):
    HTAB(X+8):PRINT "*"
110 VTAB(3):HTAB(X):PRINT "□"
```



Page 21
Adding a railway track
```
25 VTAB(9):HTAB(X):PRINT "="
```

Changing direction
```
117 VTAB(3):HTAB(X+16):PRINT "□"
140 A$="":IF PEEK(-16384)>127 THEN GET A$
```

Page 22
Computer postcard
```
 10 HOME
110 HOME
```

Page 24
Birthday fact-finder
```
10 HOME:DIM N(12)
```

Page 26
Star racer
```
10 HOME
40 VTAB(K+K):HTAB(1):PRINT K
70 VTAB(K+K):HTAB(X(K)):PRINT "   *"
```

Page 27
Adding a race track
```
51 VTAB(1):HTAB(1)
```

Race game
```
28 HOME:LET Y=0
```

Apple conversions

Page 28
Obstacle course game

```
 10 LET A=39:LET B=23:LET Z$="0"
 20 HOME:DIM Z(A,B)
 40 VTAB(W):HTAB(Q):PRINT "*"
 70 A$="":IF PEEK(-16384)>127 THEN GET A$
100 LET Z(X,Y)=1
105 VTAB(Y):HTAB(X):PRINT Z$
```

Page 29
Improving the game

```
34 IF Z(K,L)=1 THEN VTAB(L):HTAB(K):PRINT Z$
```

Page 31
Find the key

```
10 HOME:LET N=0:LET A=ASC("A")
40 A$="":IF PEEK(-16384)>127 THEN GET A$
70 LET S=INT((N/56)*10)/10
```

Page 32
Guess the number

```
10 HOME
```

Page 33
Mindreader

```
 10 HOME
240 HOME
```

Page 34
Geography quiz

```
 10 HOME
180 HOME
```

Page 35
Times table quiz

```
30 HOME
```

Key chart

This chart shows the names that the different computers give to the RETURN, ESCAPE and DELETE keys. When you are told to use one of these in this book, you should press your computer's key.

Names used in this book	RETURN	ESCAPE	DELETE
BBC	RETURN	ESCAPE *	DELETE
Electron	RETURN	ESCAPE *	DELETE
VIC 20	RETURN	RUN STOP	INST DEL
Commodore 64	RETURN	RUN STOP	INST DEL
Apple	RETURN	CTRL + C (Press both these keys together.)	←
Spectrum	ENTER	CAPS SHIFT + BREAK SPACE (Press both these keys together.)	CAPS SHIFT + 0 – (Press both these keys together.)

*There is also a BREAK key, but do not use it because it wipes the program from the computer's memory.

Bug-spotting guide

If a program does not work, it usually means there is a mistake or bug in it. The computer may display an error message which tells you what sort of mistake it is and what line it is in. If not, list the program and check each line very carefully. Even tiny bugs can stop a program working. There are some hints for what to look for below.



1

```
10 CLS
```

On some computers you would need to change this line.

Make sure that you have made all the line changes needed for your computer. Check the lines on stripes against the conversions for your computer.

2

```
20 PRINT "WHAT'S YOUR NAME?"
30 INOUT N$
```

The BASIC command INPUT is wrongly spelt.

Look out for simple typing mistakes in BASIC commands. The computer will not understand if you spell these wrongly.

3

```
40 PRINT "HOW OLD ARE YOU?
50 INPUT A
```

Missing quotes

Check for missing quotation marks. It is very easy to forget these when you are typing a program and it confuses the computer.

4

```
60 PRINT N$;" IS "A
```

Missing semi-colon

Make sure that you have put in all the punctuation marks. They are very important in BASIC as they have special meanings.

5

```
70 PRINT "YEARS OLD"
```

Letter key used instead of number key.

Check that you have typed all the numbers with the number keys. It is very easy to muddle the number 0 with the letter O.

6

If you still cannot find a bug, try retyping the longer or more complicated lines again. You might correct a mistake you have not noticed.

Index

First published in 1984 by
Usborne Publishing Ltd,
20 Garrick Street,
London WC2E 9BJ,
England.

FIRST COMPUTER LIBRARY

The First Computer Library is a lively and colourful new series for young beginners to computing. A gang of friendly ghosts and monsters show the reader what computers can do, how they work and how to program them. The books are packed with short, entertaining programs which will run on most makes of home computer.

All About Computers
This colourful introduction explains in simple terms what a computer is, how it works and what it can do. Short programs are included to illustrate the explanations.



Simple BASIC
A simple and fun introduction to programming in BASIC, with masses of entertaining programs to try and puzzles to solve.

Computer Fun
This book is packed with programs to make a computer flash messages, play games, make pictures and lots besides. No previous programming knowledge is needed.

The Usborne First Guide to the BBC Micro
This friendly guide for young readers explains exactly how to set up and use a BBC computer. A group of little monsters explain how to program in BBC BASIC, give helpful advice on things that can go wrong and show what the BBC micro can do.



ISBN 0-86020-803-6

9 780860 208037

Printed in Spain

ISBN 0 86020 803 6

£1.99